$X \to \gamma$, $H \to \eta$, $T \to \tau$

$T \to X$

Meta–Groups, Meta–Bicrossed–Products, and the Alexander Polynomial, 1

Dror Bar–Natan in Regina, June 2012

http://www.math.toronto.edu/~drorbn/Talks/Regina-1206/

Abstract. The a priori expectation of first year elementary school students who were just introduced to the natural numbers, if they would be ready to verbalize it, must be that soon, perhaps by second grade, they'd master the theory and know all there is to know about those numbers. But they would be wrong, for number theory remains a thriving subject, well-connected to practically anything there is out there in mathematics.

I was a bit more sophisticated when I first heard of knot theory. My first thought was that it was either trivial or intractable, and most definitely, I wasn't going to learn it is interesting. But it is, and I was wrong, for the reader of knot theory is often lead to the most interesting and beautiful structures in topology, geometry, quantum field theory, and algebra.

Today I will talk about just one minor example, mostly having to do with the link to algebra: A straightforward proposal for a group-theoretic invariant of knots fails if one really means groups, but works once generalized to meta-groups (to be defined). We will construct one complicated but elementary meta-group as a meta-bicrossed-product (to be defined), and explain how the resulting invariant is a not-yet-understood yet potentially significant generalization of the Alexander polynomial, while at the same time being a specialization of a somewhat-understood "universal finite type invariant of w-knots" and of an elusive "universal finite type invariant of v-knots".





"divide and conquer"

Idea. Given a group G and two "YB" pairs $R^\pm = (g_o^\pm, g_u^\pm) \in G^2$, map them to xings and "multiply along", so that



$$\overset{Z}{\longmapsto} \begin{pmatrix} g_o^+ g_u^+ g_o^+ g_u^- g_o^- g_u^+ g_o^+ g_u^+ \\ g_u^- g_o^- \end{pmatrix}$$

This Fails! R2 implies that $g_o^\pm g_o^\mp = e = g_u^\pm g_u^\mp$ and then R3 implies that g_o^+ and g_u^+ commute, so the result is a simple counting invariant.



"w-generators" go here.

A Group Computer. Given G, can store group elements and perform operations on them:



…so that $m_u^{xy} /\!\!/ m_v^{uz} = m_u^{yz} /\!\!/ m_v^{xu}$ (or $m_v^{uz} \circ m_u^{xy} = m_v^{xu} \circ m_u^{yz}$, in old-speak).

Also has S_x for inversion, e_x for unit insertion, d_x for register deletion, Δ_{xy}^z for element cloning, ρ_y^x for renamings, and $(D_1, D_2) \mapsto D_1 \cup D_2$ for merging, and many obvious composition axioms relating those. $P = \{x : g_1, y : g_2\} \Rightarrow P = \{d_y P\} \cup \{d_x P\}$

A Meta-Group. Is a similar "computer", only its internal structure is unknown to us. Namely it is a collection of sets $\{G_X\}$ indexed by all finite sets X, and a collection of operations m_z^{xy}, S_x, e_x, d_x, Δ_{xy}^z (sometimes), ρ_y^x, and $\cup$, satisfying the exact same *linear* properties.

Example 1. The non-meta example, $G_X := G^X$.

Example 2. $G_X := M_{X\times X}(\mathbb{Z})$, with simultaneous row and column operations, and "block diagonal" merges.

Bicrossed Products. If $G = HT$ is a group presented as a product of two of its subgroups, with $H \cap T = \{e\}$, then also $G = TH$ and G is determined by H, T, and the "swap" map $sw^{th} : (t,h) \mapsto (h', t')$ defined by $th = h't'$. The map sw satisfies (1) and (2) below; conversely, if $sw : T \times H \to H \times T$ satisfies (1) and (2) (+ lesser conditions), then (3) defines a group structure on $H \times T$, the "bicrossed product".



$tm_1^{12} /\!\!/ sw_{14} = sw_{24} /\!\!/ sw_{14} /\!\!/ tm_1^{12}$ $\qquad$ $gm_3^{12} := sw_{12} /\!\!/ tm_3^{12} /\!\!/ hm_3^{12}$

The ordinary Alexander.

1. Quick to compute, but computation departs from topology
2. Extends to tangles, at an exponential cost.
3. Hard to categorify.

A. add boxes 1. The standard definition of Alexander.

2. The need for for an extension to tangles. "universal"

3. The existing Alexander for tangles, aside on Alexander homology, weaknesses of current

Alexander for tangles.

4. Another example of a meta-bicrossed-product.

Just ref. again to Example 2.

Meta–Groups, Meta–Bicrossed–Products, and the Alexander Polynomial, 2

A **Meta-Bicrossed-Product** is a collection of sets $\beta(H,T)$ and operations tm_z^{xy}, hm_z^{xy} and sw_{xy}^{th} (and lesser ones), such that tm and hm are "associative" and (1) and (2) hold (+ lesser conditions). A meta-bicrossed-product defines a meta-group with $G_X := \beta(X,X)$ and gm as in (3).

β Calculus. Let $\beta(H,T)$ be

$$\left\{ \begin{array}{c|ccc} \omega & h_1 & h_2 & \cdots \\ \hline t_1 & \alpha_{11} & \alpha_{12} & \cdot \\ t_2 & \alpha_{21} & \alpha_{22} & \cdot \\ \vdots & \cdot & \cdot & \cdot \end{array} \;\middle|\; \begin{array}{l} h_j \in H,\ t_i \in T, \text{ and } \omega \text{ and} \\ \text{the } \alpha_{ij} \text{ are rational func-} \\ \text{tions in a variable } T \end{array} \right\},$$

$$tm_z^{xy}: \begin{array}{c|c} \omega & \cdots \\ \hline t_x & \alpha \\ t_y & \beta \\ \vdots & \gamma \end{array} \mapsto \begin{array}{c|c} \omega & \cdots \\ \hline t_z & \alpha+\beta \\ \vdots & \gamma \end{array}, \qquad \begin{array}{c|c} \omega_1 & H_1 \\ \hline T_1 & \alpha_1 \end{array} \cup \begin{array}{c|c} \omega_2 & H_2 \\ \hline T_2 & \alpha_2 \end{array} = \begin{array}{c|cc} \omega_1\omega_2 & H_1 & H_2 \\ \hline T_1 & \alpha_1 & 0 \\ T_2 & 0 & \alpha_2 \end{array},$$

$$hm_z^{xy}: \begin{array}{c|ccc} \omega & h_x & h_y & \cdots \\ \hline \vdots & \alpha & \beta & \gamma \end{array} \mapsto \begin{array}{c|cc} \omega & h_z & \cdots \\ \hline \vdots & \alpha+\beta+\langle\alpha\rangle\beta & \gamma \end{array},$$

$$sw_{xy}^{th}: \begin{array}{c|cc} \omega & h_y & \cdots \\ \hline t_x & \alpha & \beta \\ \vdots & \gamma & \delta \end{array} \mapsto \begin{array}{c|cc} \omega\epsilon & h_y & \cdots \\ \hline t_x & \alpha(1+\langle\gamma\rangle/\epsilon) & \beta(1+\langle\gamma\rangle/\epsilon) \\ \vdots & \gamma/\epsilon & \delta-\gamma\beta/\epsilon \end{array},$$

where $\epsilon := 1+\alpha$ and $\langle c\rangle := \sum_i c_i$, and let

$$R_{xy}^p := \begin{array}{c|cc} 1 & h_x & h_y \\ \hline t_x & 0 & T-1 \\ t_y & 0 & 0 \end{array} \qquad R_{xy}^m := \begin{array}{c|cc} 1 & h_x & h_y \\ \hline t_x & 0 & T^{-1}-1 \\ t_y & 0 & 0 \end{array}.$$

Theorem. Z^β is a tangle invariant (and more). Restricted to knots, the ω part is the Alexander polynomial. On links, it contains the multivariable Alexander polynomial. On braids, it is equivalent to the Burau representation.

Why Happy? • Applications to w-knots.
• Everything that I know about the Alexander polynomial can be expressed cleanly in this language (even if without proof), except HF, but including genus, ribbonness, cabling, v-knots, knotted graphs, etc., and there's potential for vast generalizations.
• The least wasteful "Alexander for tangles" I'm aware of.
• Every step along the computation is the invariant of something.
• Fits on one sheet, including implementation.

I mean business!

```
βSimp = Factor; SetAttributes[βCollect, Listable];
βCollect[B[ω_, Λ_]] := B[βSimp[ω],
  Collect[Λ, h_, Collect[#, t_, βSimp] &]];
βForm[B[ω_, Λ_]] := Module[{ts, hs, M},
  ts = Union[Cases[B[ω, Λ], t_, ∞, Infinity]];
  hs = Union[Cases[B[ω, Λ], h_, ∞, Infinity]];
  M = Outer[βSimp[Coefficient[Λ, h# t#]] &, hs, ts];
  PrependTo[M, t# & /@ ts];
  M = Prepend[Transpose[M], Prepend[h# & /@ hs, ω]];
  MatrixForm[M]];
βForm[else_] := else /. β_B ⧴ βForm[β];
Format[β_B, StandardForm] := βForm[β];

⟨c_⟩ := c /. t_ → 1;
tm_{x_,y_→z_}[β_] := βCollect[β /. t_{x|y} → t_z];
hm_{x_,y_→z_}[B[ω_, Λ_]] := Module[
  {α = D[Λ, h_x], β = D[Λ, h_y], γ = Λ /. h_{x|y} → 0},
  B[ω, (α + (1 + ⟨α⟩) β) h_z + γ] // βCollect];
sw_{x_,y_}[B[ω_, Λ_]] := Module[{α, β, γ, δ, ε},
  α = Coefficient[Λ, h_y t_x]; β = D[Λ, t_x] /. h_y → 0;
  γ = D[Λ, h_y] /. t_x → 0;  δ = Λ /. h_y | t_x → 0;
  ε = 1 + α;
  B[ω ε, α (1 + ⟨γ⟩/ε) h_y t_x + β (1 + ⟨γ⟩/ε) t_x
     + γ/ε h_y                + δ - γ β/ε
  ] // βCollect];
gm_{x_,y_→z_}[β_] := β // sw_{xy} // hm_{xy→z} // tm_{xy→z};
B /: B[ω1_, Λ1_] B[ω2_, Λ2_] := B[ω1 ω2, Λ1 + Λ2];
Rp_{x_,y_} := B[1, (T - 1) t_x h_y];
Rm_{x_,y_} := B[1, (T^-1 - 1) t_x h_y];
```

The key trick: $\begin{array}{c|c} \omega & h_j \\ \hline t_i & \alpha_{ij} \end{array} \longleftrightarrow B(\omega, \Lambda = \sum_{i,j} \alpha_{ij} t_i h_j)$.

```
{β = B[ω, Sum[α_{10 i+j} t_i h_j, {i, {1, 2, 3}}, {j, {4, 5}}]],
 (β // tm_{12→1} // sw_{14}) == (β // sw_{24} // sw_{14} // tm_{12→1})}
```

$$\left\{ \begin{pmatrix} \omega & h_4 & h_5 \\ t_1 & \alpha_{14} & \alpha_{15} \\ t_2 & \alpha_{24} & \alpha_{25} \\ t_3 & \alpha_{34} & \alpha_{35} \end{pmatrix}, \text{True} \right\}$$



Some testing

```
{Rm_{51} Rm_{62} Rp_{34} // gm_{14→1} // gm_{25→2} // gm_{36→3},
 Rp_{61} Rm_{24} Rm_{35} // gm_{14→1} // gm_{25→2} // gm_{36→3}}
```

$$\left\{ \begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+T}{T} & 0 \\ t_3 & \frac{-1+T}{T} & -\frac{-1+T}{T} \end{pmatrix}, \begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+T}{T} & 0 \\ t_3 & \frac{-1+T}{T} & -\frac{-1+T}{T} \end{pmatrix} \right\}$$

. . . divide and conquer!

```
β = Rm_{12,1} Rm_{27} Rm_{83} Rm_{4,11} Rp_{16,5} Rp_{6,13} Rp_{14,9} Rp_{10,15}
```

8_{17}

$$\begin{pmatrix} 1 & h_1 & h_3 & h_5 & h_7 & h_9 & h_{11} & h_{13} & h_{15} \\ t_2 & 0 & 0 & 0 & -\frac{-1+T}{T} & 0 & 0 & 0 & 0 \\ t_4 & 0 & 0 & 0 & 0 & 0 & -\frac{-1+T}{T} & 0 & 0 \\ t_6 & 0 & 0 & 0 & 0 & 0 & 0 & -1+T & 0 \\ t_8 & 0 & -\frac{-1+T}{T} & 0 & 0 & 0 & 0 & 0 & 0 \\ t_{10} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+T \\ t_{12} & -\frac{-1+T}{T} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ t_{14} & 0 & 0 & 0 & 0 & -1+T & 0 & 0 & 0 \\ t_{16} & 0 & 0 & -1+T & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

```
Do[β = β // gm_{1k→1}, {k, 2, 10}]; β
```

8_{17}, cont.

$$\begin{pmatrix} \frac{1}{T} & h_1 & h_{11} & h_{13} & h_{15} \\ t_1 & -\frac{(-1+T)(1+T)}{T} & -(-1+T)(1-T+T^2) & (-1+T)(1-T+T^2) & -1+T \\ t_{12} & -\frac{-1+T}{T} & 0 & 0 & 0 \\ t_{14} & -1+T & \frac{(-1+T)^2(1-T+T^2)}{T} & -\frac{(-1+T)^2(1-T+T^2)}{T} & 0 \\ t_{16} & \frac{-1+T}{T} & (-1+T)^2 & -\frac{(-1+T)^3}{T} & 0 \end{pmatrix}$$

James Waddell Alexander

```
Do[β = β // gm_{1k→1}, {k, 11, 16}]; β
```

$$\left(-\frac{1-4T+8T^2-11T^3+8T^4-4T^5+T^6}{T^3} \right)$$

```
<< KnotTheory`
Alexander[Knot[8, 17]][T] // Factor
Loading KnotTheory` version of August 22, 2010, 13:36:57.55.
Read more at http://katlas.org/wiki/KnotTheory.
KnotTheory::loading : Loading precomputed data in PD4Knots`.
```

$$-\frac{1-4T+8T^2-11T^3+8T^4-4T^5+T^6}{T^3}$$

A Partial To Do List. 1. Where does it *more simply* come from?
2. Remove all the denominators.
3. How do determinants arise in this context?
4. Understand links.
5. Find the "reality condition".
6. Do some "Algebraic Knot Theory".
7. Categorify.
8. Do the same in other natural quotients of the v/w-story.

trivial

ribbon

example

"God created the knots, all else in topology is the work of mortals."
Leopold Kronecker (modified)

www.katlas.org

a variant contains ...

save vertical space

Meta–Groups, Meta–Bicrossed–Products, and the Alexander Polynomial, 3

Where does it come from? The accidental[1] answer is that it is a symbolic calculus for a natural reduction[4] of the unique homomorphic expansion[2] of w-tangles[3].

1. "Accidental" for it's only how I came about it. There ought to be a better answer.
2. A "homomorphic expansion", aka as a homomorphic universal finite type invariant, is a completely canonical construct whose presence implies that the objects in questions are susceptible to study using graded algebra.
3. "v-Tangles" are the meta-group generated by crossings modulo Reidemeister moves. "w-Tangles" are a natural quotient of v-tangles. They are at least related and perhaps identical to a certain class of 1D/2D knots in 4D.
4. To "only what is visible by the 2D Lie algebra".

A certain generalization will arise by not reducing as in 4. A vast generalization may arise when homomorphic expansions for v-tangles are understood, a task likely equivalent to the Etingof-Kazhdan quantization of Lie bialgebras.